



NonLinear Materials

Introduction

This tutorial was completed using ANSYS 7.0. The purpose of the tutorial is to describe how to include material nonlinearities in an ANSYS model. For instance, the case when a large force is applied resulting in stresses greater than yield strength. In such a case, a multilinear stress-strain relationship can be included which follows the stress-strain curve of the material being used. This will allow ANSYS to more accurately model the plastic deformation of the material.



For this analysis, a simple tension specimen 100 mm X 5 mm X 5 mm is constrained at the bottom and has a load pulling on the top. This specimen is made out of an experimental substance called "WhoKilledKenium". The stress-strain curve for the substance is shown above. Note the linear section up to approximately 225 MPa where the Young's Modulus is constant (75 GPa). The material then begins to yield and the relationship becomes plastic and nonlinear.

Preprocessing: Defining the Problem

1. Give example a Title

Utility Menu > File > Change Title ...
/title, NonLinear Materials

2. Create Keypoints

Preprocessor > Modeling > Create > Keypoints > In Active CS
/PREP7
K, #, X, Y

We are going to define 2 keypoints (the beam vertices) for this structure to create a beam with a length of 100 millimeters:

Keypoint	Coordinates (x,y)
1	(0,0)
2	(0,100)

3. Define Lines

Preprocessor > Modeling > Create > Lines > Lines > Straight Line
Create a line between Keypoint 1 and Keypoint 2.
L, 1, 2

4. Define Element Types

Preprocessor > Element Type > Add/Edit/Delete...

For this problem we will use the LINK1 (2D spar) element. This element has 2 degrees of freedom (translation along the X and Y axis's) and can only be used in 2D analysis.

5. Define Real Constants

Preprocessor > Real Constants... > Add...

In the 'Real Constants for LINK1' window, enter the following geometric properties:

- i. Cross-sectional area AREA: 25
- ii. Initial Strain: 0

This defines an element with a solid rectangular cross section 5 x 5 millimeters.

6. Define Element Material Properties

Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic


In the window that appears, enter the following geometric properties for steel:

- i. Young's modulus EX: 75e3
- ii. Poisson's Ratio PRXY: 0.3

Now that the initial properties of the material have been outlined, the stress-strain data must be included.


Preprocessor > Material Props > Material Models > Structural > Nonlinear > Elastic > Multilinear Elastic

The following window will pop up.



Fill in the STRAIN and STRESS boxes with the following data. These are points from the stress-strain curve shown above, approximating the curve with linear interpolation between the points. When the data for the first point is input, click Add Point to add another. When all the points have been inputted, click Graph to see the curve. It should look like the one shown above. Then click OK.

Curve Points	Strain	Stress
1	0	0
2	0.001	75
3	0.002	150
4	0.003	225
5	0.004	240
6	0.005	250
7	0.025	300
8	0.060	355
9	0.100	390
10	0.150	420



11	0.200	435
12	0.250	449
13	0.275	450

To get the problem geometry back, select **Utility Menu > Plot > Replot.**
/REPLOT

7. Define Mesh Size

Preprocessor > Meshing > Manual Size > Size Cntrls > Lines > All Lines...

For this example we will specify an element edge length of 5 mm (20 element divisions along the line).

8. Mesh the frame

Preprocessor > Meshing > Mesh > Lines > click 'Pick All'
LMESH, ALL

Solution: Assigning Loads and Solving


1. Define Analysis Type

Solution > New Analysis > Static
ANTYPE, 0

2. Set Solution Controls

- o Select **Solution > Analysis Type > Sol'n Control...**

The following image will appear:



Ensure the following selections are made under the 'Basic' tab (as shown above)


- Ensure Large Static Displacements are permitted (this will include the effects of large deflection in the results)
- Ensure Automatic time stepping is on. Automatic time stepping allows ANSYS to determine appropriate sizes to break the load steps into. Decreasing the step size usually ensures better accuracy, however, this takes time. The Automatic Time Step feature will determine an appropriate balance. This feature also activates the ANSYS bisection feature which will allow recovery if convergence fails.
- Enter 20 as the number of substeps. This will set the initial substep to $1/20^{\text{th}}$ of the total load.
- Enter a maximum number of substeps of 1000. This stops the program if the solution does not converge after 1000 steps.
- Enter a minimum number of substeps of 1.

F. Ensure all solution items are written to a results file. This means rather than just recording the data for the last load step, data for every load step is written to the database. Therefore, you can plot certain parameters over time.

Ensure the following selection is made under the 'Nonlinear' tab (as shown below)

G. Ensure Line Search is 'On'. This option is used to help the Newton-Raphson solver converge.

H. Ensure Maximum Number of Iterations is set to 1000



NOTE

There are several options which have not been changed from their default values. For more information about these commands, type `help` followed by the command into the command line.

3. Apply Constraints

Solution > Define Loads > Apply > Structural > Displacement > On Keypoints
Fix Keypoint 1 (ie all DOFs constrained).


4. Apply Loads

Solution > Define Loads > Apply > Structural > Force/Moment > On Keypoints
Place a 10,000 N load in the FY direction on the top of the beam (Keypoint 2).

5. Solve the System

Solution > Solve > Current LS
SOLVE


The following will appear on your screen for NonLinear Analyses



This shows the convergence of the solution.


General Postprocessing: Viewing the Results

- To view the element in 2D rather than a line: **Utility Menu > PlotCtrls > Style > Size and Shape** and turn 'Display of element' ON (as shown below).



- View the deflection contour plot**

General Postproc > Plot Results > Contour Plot > Nodal Solu... > DOF solution, UY
PLNSOL, U, Y, 0, 1




Other results can be obtained as shown in previous linear static analyses.

Time History Postprocessing: Viewing the Results


As shown, you can obtain the results (such as deflection, stress and bending moment diagrams) the same way you did in previous examples using the General Postprocessor. However, you may wish to view time history results such as the deflection of the object over time.


1. Define Variables

- Select: **Main Menu > TimeHist Postpro**. The following window should open automatically.




If it does not open automatically, select **Main Menu > TimeHist Postpro > Variable Viewer**


- Click the add button  in the upper left corner of the window to add a variable.
- Select **Nodal Solution > DOF Solution > Y-Component of displacement** (as shown below) and click OK. Pick the uppermost node on the beam and click OK in the 'Node for Data' window.




- To add another variable, click the add button again. This time select **Reaction Forces > Structural Forces > Y-Component of Force**. Pick the lowermost node on the beam and click OK.
- On the Time History Variable window, click the circle in the 'X-Axis' column for FY_3. This will make the reaction force the x-variable. The Time History Variables window should now look like this:



2. Graph Results over Time

- Click on UY_2 in the Time History Variables window.
- Click the graphing button  in the Time History Variables window.
- The labels on the plot are not updated by ANSYS, so you must change them manually. Select **Utility Menu > PlotCtrls > Style > Graphs > Modify Axes** and re-label the X and Y-axis appropriately.



This plot shows how the beam deflected linearly when the force, and subsequently the stress, was low (in the linear range). However, as the force increased, the deflection (proportional to strain) began to increase at a greater rate. This is because the stress in the beam is in the plastic range and thus no longer relates to strain linearly. When you verify this example analytically, you will see the solutions are very similar. The difference can be attributed to the ANSYS solver including large deflection calculations.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to '**File > Read input from...**' and select the file.

```
finish
/clear

/prep7                ! Enter Preprocessor
k,1,0,0              ! Keypoints
k,2,0,100
l,1,2                ! Line connecting keypoints

ET,1,LINK1          ! Element type
R,1,25              ! Area of 25

MP,EX,1,75000       ! Young's modulus
MP,PRXY,1,0.3       ! Poisson's ratio

TB,MELA,1,1,12,     ! Create a table of 12 data points
                    ! to map the stress-strain curve

TBPT,,.001,75       ! Data points
TBPT,,.002,150
TBPT,,.003,225
TBPT,,.004,240
TBPT,,.005,250
```



```

TBPT,,,025,300
TBPT,,,06,355
TBPT,,,1,390
TBPT,,,15,420
TBPT,,,2,435
TBPT,,,25,449
TBPT,,,275,450

ESIZE,5           ! Element size 5
LMESH,all         ! Line mesh all lines

FINISH
/SOLU             ! Enter solution phase

NLGEOM,ON         ! Nonlinear geometry on
NSUBST,20,1000,1 ! 20 load steps
OUTRES,ALL,ALL    ! Output data for all load steps
AUTOTS,ON         ! Auto time-search on
LNSRCH,ON         ! Line search on
NEQIT,1000        ! 1000 iteration maximum

ANTYPE,0          ! Static analysis
DK,1,all          ! Constrain keypoint 1
FK,2,FY,10000     ! Load on keypoint 2

SOLVE
FINISH

/POST1            ! Enter post processor
/ESHAPE,1         ! Show element shape
PLNSOL,U,Y,0,1    ! Plot deflection contour

FINISH

/POST26           ! Enter time history
RFORCE,2,1,F,Y    ! Reads force data in variable 2
NSOL,3,2,U,Y      ! Reads y-deflection data into var 3
XVAR,2            ! Make variable 2 the x-axis
PLVAR,3
/AXLAB,Y,DEFLECTION ! Changes y label
/AXLAB,X,LOAD      ! Changes X label
/REPLOT

```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>